

Generic Programming with Fixed Points for Mutually Recursive Datatypes

Alexey Rodriguez Yakushev¹ Stefan Holdermans² Andres Löh² Johan Jeuring^{2,3}

¹Vector Fabrics B.V., Paradijslaan 28, 5611 KN Eindhoven, The Netherlands

²Department of Information and Computing Sciences, Utrecht University, P.O. Box 80.089, 3508 TB Utrecht, The Netherlands

³School of Computer Science, Open University of the Netherlands, P.O. Box 2960, 6401 DL Heerlen, The Netherlands

alexey.rodriguez@gmail.com {stefan,andres,johanj}@cs.uu.nl

Abstract

Many datatype-generic functions need access to the recursive positions in the structure of the datatype, and therefore adopt a fixed point view on datatypes. Examples include variants of fold that traverse the data following the recursive structure, or the Zipper data structure that enables navigation along the recursive positions. However, Hindley-Milner-inspired type systems with algebraic datatypes make it difficult to express fixed points for anything but regular datatypes. Many real-life examples such as abstract syntax trees are in fact systems of mutually recursive datatypes and therefore excluded. Using Haskell's GADTs and type families, we describe a technique that allows a fixed-point view for systems of mutually recursive datatypes. We demonstrate that our approach is widely applicable by giving several examples of generic functions for this view, most prominently the Zipper.

Categories and Subject Descriptors D.1.1 [Programming Techniques]: Applicative (Functional) Programming; D.2.13 [Software Engineering]: Reusable Software—Reusable libraries; D.3.3 [Programming Languages]: Language Constructs and Features—Data types and structures

General Terms Design, Languages

1. Introduction

One of the most ubiquitous activities in software development is *structuring data*. Indeed, many programming methods and development tools center around the creation of datatypes (or XML schemas, UML models, classes, grammars, et cetera). Once the structure of the data has been decided on, a programmer adds *functionality* to the datatypes. Here, there is always some functionality that is specific to a datatype – and part of the reason that the datatype has been designed in the first place. Other functionality is, however, *generic* and similar or even the same on many datatypes. Classic examples of such generic functionality are testing for equality, ordering, parsing, and pretty printing.

Implementing generic functionality can be tiresome and, therefore, error-prone: it involves adapting and applying the same high-

level programming patterns to different datatypes, time and time again. *Datatype-generic programming* alleviates this burden by enabling programmers to write *generic functions*, i.e., functions that are defined once, but that can be used on many different datatypes.

Over the years, a vast body of work has emerged on adding support for datatype-generic programming to mainstream functional programming languages, most notably Haskell (Peyton Jones 2003). While early proposals encompassed extending the underlying language with dedicated new constructs for generic programming (Jansson and Jeuring 1997; Hinze 2000a,b), recent approaches favour the definition of generic functions in Haskell itself using Haskell's advanced type-class system (Cheney and Hinze 2002; Hinze 2004; Lämmel and Peyton Jones 2003).

The various approaches to generic programming generally differ in the expressivity of the generic functions that can be defined and the classes of datatypes that are supported. The most prominent example is that quite a number of generic functions operate on the recursive structure of datatypes, but most approaches do not provide access to the recursive positions in a datatype's definition. The approaches that do provide access to these recursive positions are limited in the sense that they only apply to a restricted set of datatypes. In particular, the full recursive structure of families of mutually recursive datatypes is beyond the reach of these approaches. Still, many real-life applications of functional programming do involve mutually recursive datatypes, arguably the most striking example being the representation of abstract syntax trees in compilers. Moreover, the generic functions that arise in such applications typically require access to the full recursive structure of these types; examples include navigation (Huet 1997; Hinze et al. 2004; McBride 2008), unification (Jansson and Jeuring 1998), rewriting (Jansson and Jeuring 2000; Van Noort et al. 2008), and pattern matching (Jeuring 1995) and, more generally, recursion schemes such as *fold* and the like (Meijer et al. 1991) and downwards accumulations (Gibbons 2000).

In this paper, we present an approach to datatype-generic programming embedded in Haskell that does enable the definition of generic functions over the full recursive structure of mutually recursive datatypes. Specifically, our contributions are the following:

- We show how to generalise the encoding of regular datatypes as fixed points of functors (reviewed in Section 2) to arbitrary families of mutually recursive types. We make use of a single higher-order fixed point operator (Section 3).
- The functors for families of mutually recursive datatypes can be constructed from a small set of combinators, thereby enabling datatype-generic programming (Section 4).

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. To copy otherwise, to republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee.

ICFP'09, August 31–September 2, 2009, Edinburgh, Scotland, UK.
Copyright © 2009 ACM 978-1-60558-332-7/09/08...\$5.00

- We present several applications of generic programming in this setting, most notably the Zipper for mutually recursive types in Section 5 and generic rewriting in Section 6.

Related work is presented in Section 7, and future work and conclusions in Section 8.

A strength of our approach is that it can be readily implemented in Haskell, making use of language extensions such as type families (Schrijvers et al. 2008) and GADTs (Peyton Jones et al. 2006). The `multirec` and `zipper` libraries that are based on this paper can be obtained from HackageDB.

2. Fixed points for representing regular datatypes

Let us first review generic programming using fixed points for regular datatypes. While this is well-known, it serves not only as an introduction to the terminology we are using, but also as a template for our introduction of the more general case for families of mutually recursive types in Section 4.

A functor is a type constructor of kind $* \rightarrow *$ for which we can define a *map* function. Fixed points are represented by an instance of the `Fix` datatype:

```
data Fix f = In { out :: (f (Fix f)) }
```

Haskell’s record notation is used to introduce the selector function `out :: Fix f → f (Fix f)`.

2.1 Defining a pattern functor directly

Using `Fix`, we can represent the following datatype for simple arithmetic expressions

```
data Expr = Const Int | Add Expr Expr | Mul Expr Expr
```

by its *pattern functor*:

```
data PFExpr r = ConstF Int | AddF r r | MulF r r
type Expr' = Fix PFExpr
```

The types `Expr` and `Expr'` are isomorphic, and in Haskell we can witness this isomorphism by instantiating a class

```
class Regular a where
  deepFrom :: a → Fix (PF a)
  deepTo   :: Fix (PF a) → a
```

where

```
type family PF a :: * → *
```

The type family `PF` is an open type-level function mapping a regular type `a` of kind $* \rightarrow *$ to its pattern functor `PF a` of kind $* \rightarrow *$. We can instantiate it by saying

```
type instance PF Expr = PFExpr
```

The functions `deepFrom` and `deepTo` are straightforward to define. In practice, converting between a datatype and its fixed-point representation occurs often when programming generically, and traversing the whole value as required by `deepFrom` and `deepTo` is often more work than is actually required.

We therefore present an alternative correspondence, making use of the isomorphism

$$a \cong \text{Fix} (PF\ a) \cong (PF\ a) (\text{Fix} (PF\ a)) \cong (PF\ a)\ a$$

This means that we relate `a` to its one-layer unfolding `PF a a` (a *shallow* conversion). We redefine class `Regular` to use the following conversion functions `from` and `to`:

```
class Regular a where
  from :: a → PF a a
  to   :: PF a a → a
```

As before, in the instance for expressions,

```
instance Regular Expr where
  from = fromExpr
  to   = toExpr
```

the shallow conversion functions `fromExpr` and `toExpr` are trivial to define.

In order to establish that `PFExpr` really is a functor, we make it an instance of class `Functor`:

```
instance Functor PFExpr where
  fmap f (ConstF i) = ConstF i
  fmap f (AddF e e') = AddF (f e) (f e')
  fmap f (MulF e e') = MulF (f e) (f e')
```

Given *fmap*, many recursion schemes can be defined, for example:

```
fold :: (Regular a, Functor (PF a)) =>
  (PF a r → r) → (a → r)
fold f = f ∘ fmap (fold f) ∘ from
unfold :: (Regular a, Functor (PF a)) =>
  (r → PF a r) → (r → a)
unfold f = to ∘ fmap (unfold f) ∘ f
```

Note how the conversions in the class `Regular` allow us to work with the original datatype `Expr` rather than its fixed point representation `Expr'`, but it is easy to define the deep conversion functions. For instance, `fold In` turns a regular datatype `a` into its fixed-point representation `Fix (PF a)`.

Another recursion scheme we can define is *compos* (Bringert and Ranta 2006). Much like *fold*, it traverses a data structure and performs operations on the children. There are different variants of *compos*, the simplest is equivalent to PolyP’s *mapChildren* (Jansson and Jeuring 1998): it applies a function of type `a → a` to all children. This parameter is also responsible for performing the recursive call, because *compos* itself is not recursive:

```
compos :: (Regular a, Functor (PF a)) => (a → a) → a → a
compos f = to ∘ fmap f ∘ from
```

2.2 Building functors systematically

The approach presented above still requires us to write *fmap* by hand for every datatype. Furthermore, other applications such as navigation or rewriting require functions defined on the pattern functor that cannot directly be derived from *fmap*. Thus, having to write *fmap* and such other functions manually for each datatype is undesirable. Fortunately, it is also unnecessary.

In the following, we present a fixed set of datatypes that can be used to construct pattern functors systematically:

```
data K a      r = K a
data I r      r = I r
data (f ∶× g) r = f r ∶× g r
data (f ∶+ g) r = L (f r) | R (g r)
infixr 7 ∶×
infixr 6 ∶+
```

The type `K` is used to represent occurrences of constant types, such as `Int` and `Bool` in the `Expr` example. The type `I` represents recursive positions. Using `∶×`, we can combine different fields of a constructor, and with `∶+`, we can combine constructors.

Using the above datatypes, we can thus represent the pattern functor of `Expr` as follows:

```
type PFExpr = K Int ∶+ (I ∶× I) ∶+ (I ∶× I)
type Expr' = Fix PFExpr
```

Datatypes, such as `Expr`, whose recursive structure can be represented by a polynomial functor (consisting of sums, products and

constants) are often called *regular datatypes*. The uniform encoding allows us to define functions that work on all regular datatypes. In particular, we can now define a generic *map* function by declaring the following instances of the class `Functor`:

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
instance Functor I where
  fmap f (I x) = I (f x)
instance Functor (K a) where
  fmap _ (K x) = K x
instance (Functor f, Functor g) => Functor (f :+: g) where
  fmap f (L x) = L (fmap f x)
  fmap f (R y) = R (fmap f y)
instance (Functor f, Functor g) => Functor (f :x: g) where
  fmap f (x :x: y) = fmap f x :x: fmap f y
```

With these declarations, we obtain *fmap* on PF_{Expr} for free. Similarly, we get *fmap* on all datatypes as long as we express them as fixed points of pattern functors using `K`, `I`, `:x:` and `:+:`. Using *fmap*, we get *fold*, *unfold* and *compos* for free on all these datatypes. By providing one structural representation of a datatype (the instantiations of `PF` and `Regular`), we gain access to a multitude of powerful functions, and can easily define more.

Being able to convert between the original datatype such as `Expr` and the fixed point `Expr'` or the one-layer unfolding `PF Expr Expr` now becomes much more important, because for application-specific, non-generic functions, we want to be able to use the original constructor names rather than sequences of constructor applications for the representation. This is reflected in the fact that the conversion functions *from_{Expr}* and *to_{Expr}*, while still being entirely straightforward, now become more verbose. Here is *from_{Expr}* as an example:

```
from_Expr :: Expr -> PF Expr Expr
from_Expr (Const i) = L (K i)
from_Expr (Add e e') = R (L (I e :x: I e'))
from_Expr (Mul e e') = R (R (I e :x: I e'))
```

To facilitate the conversion, some generic programming languages automatically generate mappings that relate datatypes such as `Expr` with their structure representation counterparts (`Expr'`) (Jansson and Jeuring 1997; Löh 2004; Holdermans et al. 2006). In this case – if we do not want to extend the compiler – we can use a meta-programming tool such as Template Haskell (Sheard and Peyton Jones 2002) to generate the `PF` and `Regular` instances for a datatype.

3. Fixed points for mutually recursive datatypes

In Section 2, we have shown how we can generically program with regular datatypes by expressing them as fixed points of functors. In practice, one often has to deal with large families of mutually recursive datatypes, which are not regular. As an example, consider the following extended version of our `Expr` datatype:

```
data Expr = Const Int
          | Add Expr Expr
          | Mul Expr Expr
          | EVar Var
          | Let Decl Expr
data Decl = Var := Expr
          | Seq Decl Decl
type Var = String
```

We now have two datatypes that are mutually recursive, `Expr` and `Decl`. Both make use of a third type `Var`. In order to deal

with families such as this representation of an abstract syntax tree, we will now demonstrate how to generalize the representation of datatypes as fixed points of functors to such families.

3.1 Fixed points for a specific number of datatypes

Swierstra et al. (1999) have shown how to represent a family of two mutually recursive types as a fixed point in Haskell. The idea is to introduce a different fixed point datatype that abstracts over bifunctors of kind $* \rightarrow * \rightarrow *$ rather than functors of kind $* \rightarrow *$:

```
data Fix2 f g = In2 (f (Fix2 f g) (Fix2 g f))
```

We can easily generalize this idea further and define `Fix3`, `Fix4` and so on. Depending on whether we want to count `Var` as a full member of our abstract syntax tree family or not, we can then use either `Fix2` or `Fix3` to represent such a family as a fixed point of functors.

The problem, however, is that we can also no longer use `I`, `K`, `:+:` and `:x:` to construct functors for arbitrary families systematically. Instead, it turns out that we require new variants of `I`, `K`, `:+:` and `:x:` for each arity. In the end, we have to rework our entire generic programming machinery for each arity of family we want to support, defeating the very purpose of generic programming. Furthermore, families of datatypes can be very large, and we cannot hope that supporting a limited amount of arities will suffice in practice.

3.2 A uniform way to represent fixed points

At first, it looks like we cannot easily abstract over the arities of fixed-points in Haskell. However, it is well-known that an n -tuple of types can be encoded as a function taking an index (between 0 and $n - 1$) to a type. In other words, we have an isomorphism between the kinds $*^n$ and $\mathbf{n} \rightarrow *$ provided that \mathbf{n} is a kind with exactly n different inhabitants.

Let us apply this idea to fixed points. A fixed point of a single datatype is given by

```
Fix :: (* -> *) -> *
```

For a family of two datatypes, we need two applications of

```
Fix2 :: (* -> * -> *) -> (* -> * -> *) -> *
```

which modulo currying is the same as

```
Fix2 :: (*2 -> *)2 -> *
```

For a family of n datatypes, we need n applications of

```
Fixn :: (*n -> *)n -> *
```

or a single application of

```
Fixn :: ((*n -> *)n -> *)n
```

Applying the isomorphism between $*^n$ and $\mathbf{n} \rightarrow *$, we get

```
Fixn :: n -> (n -> ((n -> *) -> *)) -> *
```

Reordering the arguments reveals that we have really generalized from a fixed point for kind $*$ to a fixed point for $\mathbf{n} \rightarrow *$:

```
Fixn :: ((n -> *) -> (n -> *)) -> (n -> *)
```

Apart from the fact that \mathbf{n} is not available in Haskell's kind system, we now have a uniform representation of a fixed-point combinator that is suitable to express arbitrary families of datatypes. Fortunately, the remaining gap is easy to bridge, as we show in the next section.

4. Indexed fixed points in Haskell

After having presented the idea of how to get a uniform representation of fixed points, we are now going to explain how to make use

of this idea in Haskell. We develop a library for generic programming with families of mutually recursive types much in the same style as we did in Section 2 for regular datatypes. We are going to use the family of abstract syntax trees from the introduction of Section 3 as our running example.

4.1 Encoding indexed fixed points in Haskell’s kind system

First, we have to find a way to encode \mathbf{n} in Haskell’s kind system, where \mathbf{n} is supposed to be a kind that has exactly n types as inhabitants. Haskell offers just one base kind, namely $*$, so we are left with little choice. However, we can simply approximate \mathbf{n} by $*$ in Haskell, as long as we promise to instantiate $*$ with only n different types.

In practice, if we have a family φ with n different types, we use the types in the family themselves as the indices to instantiate such positions of $*$. In this paper, we will write $*_{\varphi}$ rather than $*$ for such positions in order to make it more explicit that we are using a virtual subkind of $*$ that only consists of the members of family φ . Thus, our uniform fixed-point combinator now has kind

$$\text{HFix} :: ((*_{\varphi} \rightarrow *) \rightarrow (*_{\varphi} \rightarrow *)) \rightarrow (*_{\varphi} \rightarrow *)$$

and can be defined in Haskell as

```
data HFix (f :: (*φ → *) → (*φ → *)) (ix :: *φ) =
  HIn (f (HFix f) ix)
```

In our abstract syntax tree example, we have a family that we choose to call AST with three different types, and we are going to write $*_{\text{AST}}$ for the subkind of $*$ consisting only of the types Expr, Decl, and Var.

We go even further and introduce a family-specific GADT (that we also call φ) and define it such that a value of φ ix can serve as a proof that ix is a type that belongs to φ . Whenever we quantify over a variable of kind $*_{\varphi}$, we will pass such a value of type φ ix to make explicit that we quantify over a limited set of types.

For the example, we introduce the GADT

```
data AST :: *AST → * where
  Expr :: AST Expr
  Decl :: AST Decl
  Var  :: AST Var
```

such that a value of AST ix serves as a proof that ix is a member of the AST family.

One example where we make use of explicit proofs is when defining a *map* function for higher-order functors. Since the type has changed, we have to define a new class

```
class HFunctor (φ :: *φ → *)
  (f :: (*φ → *) → (*φ → *)) where
  hmap :: (∀ ix. φ ix → r ix → r' ix) → φ ix → f r ix → f r' ix
```

The function *hmap* has a rank-2 type. The function that is mapped is quantified over all members ix of family φ . If for every index ix in φ , this function transforms an r ix into an r' ix, then we can transform a functor with recursive calls given by r into a functor with recursive calls given by r' .

It is perhaps instructive to note that if φ is a family consisting of only one type, there will be only one choice for φ ix, and the type of *hmap* reduces to the type of *fmap* for regular functors.

Instead of using explicit proofs of type φ ix, it is sometimes helpful to use a type class

```
class El (φ :: *φ → *) (ix :: *φ) where
  proof :: φ ix
```

and then use an implicit class constraint El φ ix instead of a value of type φ ix.

For the AST family, we define the following instances:

```
instance El AST Expr where proof = Expr
instance El AST Decl where proof = Decl
instance El AST Var  where proof = Var
```

4.2 Defining a pattern functor directly

Before we discuss how to represent functors of families generically, let us show how we can represent our family for abstract syntax trees as a fixed point in terms of HFix directly.

The functor for AST can be defined as follows:

```
data PFAST :: (*AST → *) → (*AST → *) where
  ConstF :: Int      → PFAST r Expr
  AddF   :: r Expr → r Expr → PFAST r Expr
  MulF   :: r Expr → r Expr → PFAST r Expr
  EVarF  :: r Var   → PFAST r Expr
  LetF   :: r Decl → r Expr → PFAST r Expr
  BindF  :: r Var  → r Expr → PFAST r Decl
  SeqF   :: r Decl → r Decl → PFAST r Decl
  VF     :: String → PFAST r Var
```

The parameter r is used to denote a recursive call. At each recursive position, we apply r to the appropriate index in order to indicate the type we recurse on. Furthermore, each constructor of the functor targets a specific member of the family.

By using HFix on the pattern functor, we obtain types that are isomorphic to the original family:

```
type Expr' = HFix PFAST Expr
type Decl' = HFix PFAST Decl
type Var'  = HFix PFAST Var
```

The isomorphisms can be witnessed by conversion functions once more, and for this purpose, we declare a class Family that corresponds to Regular:

```
class Family φ where
  from :: φ ix → ix → PF φ I* ix
  to   :: φ ix → PF φ I* ix → ix
type family PF (φ :: *φ → *) :: (*φ → *) → (*φ → *)
```

Like in the class Regular, we decide to implement a shallow conversion rather than a deep conversion. Note that all conversion functions take a φ ix as first argument, as proof that ix is indeed a member of φ . In the pattern functor, we have to describe the type of the recursive positions by means of a datatype of kind $*_{\varphi} \rightarrow *$. The one-layer unfolding uses the original datatypes of the family in the recursive positions, and we express this by choosing I_* :

```
data I* (ix :: *φ) = I* { unI* :: ix }
```

The type I_* behaves as the identity on types so that recursive occurrences inside the functor are stored “as is”. Although the definition of I_* is essentially the same as that of I in Section 2, we give it a different name to highlight that we are using it conceptually at kind $*_{\varphi} \rightarrow *$ rather than kind $*$ $\rightarrow *$, even though the two kinds coincide in the Haskell code.

Here is the Family instance of AST:

```
type instance PF AST = PFAST
instance Family AST where
  from = fromAST
  to   = toAST
```

The functions *from*_{AST} and *to*_{AST} are straightforward and not given here.

We can now go on to define a HFunctor instance and subsequently recursion schemes such as *fold* and *unfold* for PFAST. However, since we strive for programming generically with families of datatypes, we want to avoid having to define HFunctor man-

ually for our family. Instead, we will try – as we have before in Section 2 – to build our functor systematically from a fixed set of datatypes.

4.3 Building functors systematically

It turns out that we can use almost the same datatypes as before to represent functors. The datatypes K , $:\times:$, and $:\text{+}:$ can be lifted from being parameterized over an r of kind $*$ to being parameterized over an r of kind $*\varphi \rightarrow *$ and an index ix of kind $*\varphi$:

```

data K a (r :: * $\varphi$   $\rightarrow$  *) (ix :: * $\varphi$ ) = K a
data (f :+ : g) (r :: * $\varphi$   $\rightarrow$  *) (ix :: * $\varphi$ ) = L (f r ix) | R (g r ix)
data (f : $\times$  : g) (r :: * $\varphi$   $\rightarrow$  *) (ix :: * $\varphi$ ) = f r ix : $\times$  : g r ix

```

The type I has been used to represent a recursive call. In the current situation, recursive calls can be to a specific index in the family. Therefore, I gets an additional argument $xi :: *\varphi$ that is used to determine the recursive call to make:

```

data I (xi :: * $\varphi$ ) (r :: * $\varphi$   $\rightarrow$  *) (ix :: * $\varphi$ ) = I (r xi)

```

It is perhaps surprising that xi is different from ix . But where ix projects out a certain member of the family, the type of the recursive call is independent of the type we are ultimately interested in. In fact, we have not yet used the parameter ix anywhere. If we look at the direct definition of PF_{AST} , we see that depending on the index we choose to project out of the functor, we get different functors. Only the first five constructors of PF_{AST} contribute to $PF_{AST} r Expr$, for example.

We introduce another combinator for pattern functors in order to express such constraints on the index:

```

infix 6 >:
data (f >: (xi :: * $\varphi$ )) (r :: * $\varphi$   $\rightarrow$  *) (ix :: * $\varphi$ ) where
  Tag :: f r xi  $\rightarrow$  (f >: xi) r xi

```

By tagging a functor with an index from the family, we make explicit that the tagged part only contributes to the structure of that particular member of the family.

We now have all the combinators we need to give a structural representation of the AST pattern functor:

```

type PFAST = K Int           >: Expr :+ : -- Const
              (I Expr : $\times$  : I Expr) >: Expr :+ : -- Add
              (I Expr : $\times$  : I Expr) >: Expr :+ : -- Mul
              I Var           >: Expr :+ : -- EVar
              (I Decl : $\times$  : I Expr) >: Expr :+ : -- Let
              (I Var : $\times$  : I Expr) >: Decl :+ : -- :=
              (I Decl : $\times$  : I Decl) >: Decl :+ : -- Seq
              K String        >: Var      -- V

```

To match the structure of the direct definition of PF_{AST} more closely, we have chosen to tag the representation of every constructor with the index it targets. Alternatively, we could have tagged the sum of all constructors of a type just once.

If we use the structural version of PF_{AST} in the `Family` instance, we have to adapt the conversion functions. Again, these are straightforward, but lengthy. We only show $from_{AST}$:

```

fromAST :: AST ix  $\rightarrow$  ix  $\rightarrow$  PFAST I* ix
fromAST Expr (Const i) =
  L (Tag (K i))
fromAST Expr (Add e e') =
  R (L (Tag (ci e : $\times$  : ci e')))
fromAST Expr (Mul e e') =
  R (R (L (Tag (ci e : $\times$  : ci e'))))
fromAST Expr (EVar x) =
  R (R (R (L (Tag (ci x))))))
fromAST Expr (Let d e) =
  R (R (R (R (L (Tag (ci d : $\times$  : ci e))))))

```

```

fromAST Decl (x := e) =
  R (R (R (R (R (L (Tag (ci x : $\times$  : ci e))))))))
fromAST Decl (Seq d d') =
  R (R (R (R (R (L (Tag (ci d : $\times$  : ci d'))))))))
fromAST Var x =
  R (R (R (R (R (R (Tag (K x)))))))
ci x = I (I* x)

```

4.4 Generic hmap

We still have to establish that our new functor combinators are actually higher-order functors themselves:

```

instance El  $\varphi$  xi  $\Rightarrow$  HFunctor  $\varphi$  (I xi) where
  hmap f _ (I x) = I (f proof x)
instance HFunctor  $\varphi$  (K a) where
  hmap f _ (K x) = K x
instance (HFunctor  $\varphi$  f, HFunctor  $\varphi$  g)  $\Rightarrow$ 
  HFunctor  $\varphi$  (f :+ : g) where
  hmap f p (L x) = L (hmap f p x)
  hmap f p (R y) = R (hmap f p y)
instance (HFunctor  $\varphi$  f, HFunctor  $\varphi$  g)  $\Rightarrow$ 
  HFunctor  $\varphi$  (f : $\times$  : g) where
  hmap f p (x : $\times$  : y) = hmap f p x : $\times$  : hmap f p y
instance HFunctor  $\varphi$  f  $\Rightarrow$  HFunctor  $\varphi$  (f >: ix) where
  hmap f p (Tag x) = Tag (hmap f p x)

```

Despite our generalization, the code for $hmap$ looks almost completely identical to the code for $fmap$. We need an additional, but trivial case for $(>:)$. A slight change occurs in the case for I , where we additionally have to require that the recursive call is actually in our family via $El \varphi xi$, to be able to pass the required *proof* to the mapped function f .

4.5 Generic compos

Using $hmap$, it is easy to define $compos$:

```

compos :: (Family  $\varphi$ , HFunctor  $\varphi$  (PF  $\varphi$ ))  $\Rightarrow$ 
  ( $\forall$  ix.  $\varphi$  ix  $\rightarrow$  ix  $\rightarrow$  ix)  $\rightarrow$   $\varphi$  ix  $\rightarrow$  ix  $\rightarrow$  ix
compos f p = to p  $\circ$  hmap ( $\lambda p \rightarrow I* \circ f p \circ unI*) p \circ from p$ 
```

The only differences to the version in Section 2 are due to the presence of explicit proof terms of type φix and because the actual values in the structure are now wrapped in applications of the I_* constructor.

Bringert and Ranta (2006) describe in their paper on *compos* how to define the function on families of mutually recursive datatypes. Their solution, however, requires to modify the family of datatypes and rewrite them as a single GADT. Our version of *compos* works on families of mutually recursive datatypes without modification.

As an example use of *compos*, consider the following expression:

```

example = Let ("x" := Mul (Const 6) (Const 9))
          (Add (EVar "x") (EVar "y"))

```

The following function renames all variables in *example* – note how *renameVar'* can use the type representation to take different actions for different nodes – in this case, filter out nodes of type `Var`.

```

renameVar :: Expr  $\rightarrow$  Expr
renameVar = renameVar' Expr
where
  renameVar' :: AST a  $\rightarrow$  a  $\rightarrow$  a
  renameVar' Var x = x ++ "_"
  renameVar' p x = compos renameVar' p x

```

The call `renameVar example` yields:

```
Let ("x_" := Mul (Const 6) (Const 9))
    (Add (EVar "x_") (EVar "y_"))
```

4.6 Generic fold

We can also define *fold* using *hmap*. Again, the definition is very similar to the single-datatype version:

```
type Algebra  $\varphi$  r =  $\forall ix. \varphi ix \rightarrow PF \varphi r ix \rightarrow r ix$ 
fold :: (Family  $\varphi$ , HFunctor  $\varphi$  (PF  $\varphi$ ))  $\Rightarrow$ 
    Algebra  $\varphi$  r  $\rightarrow \varphi ix \rightarrow ix \rightarrow r ix$ 
fold fp = fp  $\circ$  hmap ( $\lambda p (I_* x) \rightarrow fold fp x$ ) p  $\circ$  from p
```

Using *fold* is slightly trickier than using *compos*, because we have to construct a suitable argument of type *Algebra*. This algebra argument involves a function operating on the pattern functor, which is itself a generically derived datatype. We therefore have to write a function that destructs a sum of products, where the fields in the products are wrapped by occurrences of *K* or *I*. It is much more natural to define an algebra by giving one function per constructor, with the functions taking as many arguments as there are fields, preferably even in a curried style.

This problem is not caused by having families of many datatypes. The generic programming language PolyP (Jansson and Jeuring 1997) has a special ad-hoc construct that helps in defining algebras in a convenient style. We can do better: in the following, we will define a type-indexed datatype (Hinze et al. 2004) for algebras, as a type family inductively defined over the structure of functors. We can then define algebras in a convenient style, and use them in a generic fold.

The type-indexed datatype *Alg* is defined as follows:

```
type family Alg (f :: (* $\varphi$   $\rightarrow$  *)  $\rightarrow$  * $\varphi$   $\rightarrow$  *)
    (r :: * $\varphi$   $\rightarrow$  *) (ix :: *) :: *
type instance Alg (K a) r ix = a  $\rightarrow$  r ix
type instance Alg (I xi) r ix = r xi  $\rightarrow$  r ix
type instance Alg (f :+: g) r ix = (Alg f r ix, Alg g r ix)
type instance Alg (K a : $\times$ : g) r ix = a  $\rightarrow$  Alg g r ix
type instance Alg (I xi : $\times$ : g) r ix = r xi  $\rightarrow$  Alg g r ix
type instance Alg (f : $\triangleright$ : xi) r ix = Alg f r xi
```

The definition shows how we want to define our algebras: Occurrences of *K* and *I* are unwrapped. An algebra on a sum is a pair of algebras on the components. In the product case, we make use of knowledge on how datatypes are built: products are always nested to the right, and the left components are always fields, either wrapped by *K* or *I*. Hence, we can give two cases that allow us to turn algebras on a product into curried functions. The case for tags simply recurses.

We then have to show that we can transform such a more convenient algebra into the form that *fold* expects. To this end, we define the generic function *apply*:

```
class Apply (f :: (* $\varphi$   $\rightarrow$  *)  $\rightarrow$  * $\varphi$   $\rightarrow$  *) where
    apply :: Alg f r ix  $\rightarrow$  f r ix  $\rightarrow$  r ix
instance Apply (K a) where
    apply f (K x) = f x
instance Apply (I xi) where
    apply f (I x) = f x
instance (Apply f, Apply g)  $\Rightarrow$  Apply (f :+: g) where
    apply (f, g) (L x) = apply f x
    apply (f, g) (R x) = apply g x
instance Apply g  $\Rightarrow$  Apply (K a : $\times$ : g) where
    apply f (K x : $\times$ : y) = apply (f x) y
```

```
instance Apply g  $\Rightarrow$  Apply (I xi : $\times$ : g) where
    apply f (I x : $\times$ : y) = apply (f x) y
instance Apply f  $\Rightarrow$  Apply (f : $\triangleright$ : xi) where
    apply f (Tag x) = apply f x
```

We can further facilitate the construction of algebras by defining an infix operator for pairing:

```
infixr 1 &
(&) = (,)
```

As an example, let us specify an evaluator on our abstract syntax tree types using an algebra.

Because different types in our family are mapped to different results, we need another family of datatypes for the result type of our algebra:

```
data family Value a :: *
data instance Value Expr = EV (Env  $\rightarrow$  Int)
data instance Value Decl = DV (Env  $\rightarrow$  Env)
data instance Value Var = VV Var
type Env = [(Var, Int)]
```

An environment maps variables to integers. Expressions can contain variables, we therefore interpret them as functions from environments to integers. Declarations can be seen as environment transformers. Variables evaluate to their names. We can now state the algebra:

```
evalAlg :: Algebra AST Value
evalAlg = const (apply
    (  $\lambda x$   $\rightarrow$  EV (const x) -- Const
    & ( $\lambda (EV x) (EV y) \rightarrow EV (\lambda m \rightarrow x m + y m)$ ) -- Add
    & ( $\lambda (EV x) (EV y) \rightarrow EV (\lambda m \rightarrow x m * y m)$ ) -- Mul
    & ( $\lambda (VV x) \rightarrow EV (fromJust \circ lookup x)$ ) -- EVar
    & ( $\lambda (DV e) (EV x) \rightarrow EV (\lambda m \rightarrow x (e m))$ ) -- Let
    & ( $\lambda (VV x) (EV v) \rightarrow DV (\lambda m \rightarrow (x, v m) : m)$ ) -- :=
    & ( $\lambda (DV f) (DV g) \rightarrow DV (g \circ f)$ ) -- Seq
    & ( $\lambda x \rightarrow VV x$ )) -- V
```

Testing

```
eval :: Expr  $\rightarrow$  Env  $\rightarrow$  Int
eval x = let (EV f) = fold evalAlg Expr x in f
```

in the expression *eval example* [("y", -12)] yields 42.

4.7 Summary

We have now introduced a library for generic programming on families of mutually recursive types. The library consists of the type family *PF*, the classes *Family* and *EI*, and the functor constructors *I*, *K*, *:+:*, *: \times :*, and *: \triangleright :*. Furthermore, the library contains classes and instances for a number of generic functions, such as all the *HFunctor* code, the definitions of *compos*, *fold* and *unfold*.

To use the library for a specific family a user has to do the following: define a GADT such as *AST*, instantiate the type family *PF* to the pattern functor, and construct *Family* and *EI* instances. This may still seem a significant amount of work, but all of this code is entirely straightforward and can easily be automated. In fact, we have implemented the generation of most of this boilerplate code in Template Haskell, so that only the definition of the GADT and a call to a Template Haskell function remains.

Once the library is instantiated, all generic functions that are provided by the library are available for this family without any further work.

5. The Zipper

For a tree-like datatype, the Zipper (Huet 1997) is a derived data structure that allows efficient navigation through a tree, along its

recursive nodes. At every moment, the Zipper keeps track of a *location*: a point of focus paired with a context that represents the rest of the tree. The focus can be moved up, down, left, and right.

For regular datatypes, it is well-known how to define Zippers generically (Hinze et al. 2004). In the following, we first show how to define a Zipper for a system of mutually recursive datatypes using our example of abstract syntax trees (Section 5.1). Then, in Section 5.2, we give a generic algorithm in terms of the representations introduced in Section 4.3.

5.1 Zipper for mutually recursive datatypes

We first give a non-generic presentation of the Zipper for abstract syntax trees as defined in Section 3.

A location is the current focus paired with context information. In a setting with multiple types, the type of the focus ix is not known – hence, we make it existential, and carry around a representation of type $AST\ ix$:

```
data LocAST :: *AST → * where
  Loc :: AST ix → ix → CtxAST a ix → LocAST a
```

The type Ctx_{AST} encodes context information for the focus as a path from the focus to the root of the full tree. The path is stored in a stack of context frames:

```
data CtxAST :: *AST → *AST → * where
  Nil :: CtxAST a a
  Cons :: CtxAST ix b → CtxAST a ix → CtxAST a b
```

A context stack of type $Ctx_{AST}\ a\ b$ represents a value of type a with a b -typed *hole* in it. More specifically, a stack consists of frames of type $Ctx_{AST}\ ix\ b$ that represent constructor applications that yield an ix -value with a hole of type b in it. The full tree that is represented by a location can be recovered by plugging the value in focus into the topmost context frame, plugging the resulting value into the next frame, and so on. For this to work, the target type ix of each context frame must be equal to the type of the hole in the remainder of the stack – as enforced by the type of *Cons*.

5.1.1 Contexts

A single context frame Ctx_{AST} is following the structure of the types in the AST system closely.

```
data CtxAST :: *AST → *AST → * where
  AddC1 :: Expr → CtxAST Expr Expr
  AddC2 :: Expr → CtxAST Expr Expr
  MulC1 :: Expr → CtxAST Expr Expr
  MulC2 :: Expr → CtxAST Expr Expr
  EVarC :: CtxAST Expr Var
  LetC1 :: Expr → CtxAST Expr Decl
  LetC2 :: Decl → CtxAST Expr Expr
  BindC1 :: Expr → CtxAST Decl Var
  BindC2 :: Var → CtxAST Decl Expr
  SeqC1 :: Decl → CtxAST Decl Decl
  SeqC2 :: Decl → CtxAST Decl Decl
```

The relation between Ctx_{AST} and AST becomes even more pronounced if we also look at the directly defined pattern functor PF_{AST} from Section 4.2. For every constructor in PF_{AST} , we have as many constructors in Ctx_{AST} as there are recursive positions. We can descend into a recursive position. The type of the recursive position then becomes the type of the hole, the second argument of Ctx_{AST} . The other components of the original constructor are stored in the context. As an example, consider:

```
Let :: Decl → Expr → Expr
LetF :: r Decl → r Expr → PFAST r Expr
```

We have two recursive positions. If we descend into the first, then $Decl$ is the type of the hole, while $Expr$ remains – and so we get

```
LetC1 :: Expr → CtxAST Expr Decl
```

If, however, we descend into the second position, then $Expr$ is the type of the hole with $Decl$ remaining:

```
LetC2 :: Decl → CtxAST Expr Expr
```

5.1.2 Navigation

We now define functions that move the focus, transforming a location into a new location. These functions return their result in the *Maybe* monad, because navigation may fail: we cannot move down from a leaf of the tree, up from the root, or right if there are no more siblings in that direction.

Moving down analyzes the current focus. For all constructors that do not build leaves, we descend into the leftmost child by making it the new focus, and by pushing an appropriate frame onto the context stack. For leaves, we return *Nothing*.

```
down :: LocAST ix → Maybe (LocAST ix)
down (Loc Expr (Add e e') cs) =
  Just (Loc Expr e (Cons (AddC1 e') cs))
down (Loc Expr (Mul e e') cs) =
  Just (Loc Expr e (Cons (MulC1 e') cs))
down (Loc Expr (EVar x) cs) =
  Just (Loc Var x (Cons EVarC cs))
down (Loc Expr (Let d e) cs) =
  Just (Loc Decl d (Cons (LetC1 e) cs))
down (Loc Decl (x := e) cs) =
  Just (Loc Var x (Cons (BindC1 e) cs))
down (Loc Decl (Seq d d') cs) =
  Just (Loc Decl d (Cons (SeqC1 d') cs))
down _ = Nothing
```

The function *right* succeeds for nodes that actually have a right sibling. The size of the context stack remains unchanged: we just replace its top element with a new frame.

```
right :: LocAST ix → Maybe (LocAST ix)
right (Loc _ e (Cons (AddC1 e') cs)) =
  Just (Loc Expr e' (Cons (AddC2 e) cs))
right (Loc _ e (Cons (MulC1 e') cs)) =
  Just (Loc Expr e' (Cons (MulC2 e) cs))
right (Loc _ d (Cons (LetC1 e) cs)) =
  Just (Loc Expr e (Cons (LetC2 d) cs))
right (Loc _ x (Cons (BindC1 e) cs)) =
  Just (Loc Expr e (Cons (BindC2 x) cs))
right (Loc _ d (Cons (SeqC1 d') cs)) =
  Just (Loc Decl d' (Cons (SeqC2 d) cs))
right _ = Nothing
```

The function *left* is very similar to *right*. Finally, the function *up* is applicable whenever the current focus is not the root of the tree, i.e., whenever the context stack is non-empty. We then analyze the top context frame and plug in the old focus, yielding the new focus, and retain the rest of the context. The definition is omitted for reasons of space.

5.1.3 Using the Zipper

To use the Zipper, we need functions to turn syntax trees into locations, and back again. For manipulating trees, we provide an update operation that replaces the subtree in focus.

To enter the tree, we place it into the empty context:

```
enter :: AST ix → ix → LocAST ix
enter p e = Loc p e Nil
```

To leave, we move up as far as possible and then return the expression in focus.

```
leave :: LocAST Expr → Expr
leave (Loc _ e Nil) = e
leave loc           = leave (fromJust (up loc))
```

To update the tree, we pass in a function capable of modifying the current point of focus. Because the value in focus can have different types, this function needs to be parameterized by the type representation.

```
update :: (∀ix.AST ix → ix → ix) →
         LocAST Expr → LocAST Expr
update f (Loc p x cs) = Loc p (f p x) cs
```

As an example, we modify the multiplication in

```
example = Let ("x" := Mul (Const 6) (Const 9))
          (Add (EVar "x") (EVar "y"))
```

To combine the navigation and edit operations, it is helpful to make use of flipped function composition ($\gg\gg$):: (a → b) → (b → c) → (a → c) and monadic composition ($\gg\gg$):: Monad m ⇒ (a → m b) → (b → m c) → (a → m c). The call

```
enter Expr >>> down >>> down >>> right >>> update solve >>>
leave >>> return $ example
```

with

```
solve :: AST ix → ix → ix
solve Expr _ = Const 42
solve _      x = x
```

results in

```
Just (Let ("x" := Const 42) (Add (EVar "x") (EVar "y")))
```

5.2 A generic Zipper

We now define a Zipper generically for a system of mutually recursive datatypes. We make the same steps as in the example for abstract syntax trees before.

The type definitions for locations and context stacks stay essentially the same:

```
data Loc :: (*φ → *) → *φ → * where
  Loc :: (Family φ, Zipper φ (PF φ)) ⇒
        φ ix → ix → Ctxs φ a ix → Loc φ a
data Ctxs :: (*φ → *) → *φ → *φ → * where
  Nil  :: Ctxs φ a a
  Cons :: φ ix → Ctx (PF φ) ix b → Ctxs φ a ix → Ctxs φ a b
```

Instead of a specific proof term AST ix, we now store a generic proof term φ ix for an arbitrary family in a location. Additionally, we need a Zipper for the system φ. This condition is expressed by Zipper (PF φ) and will be explained in more detail below.

In the stack Ctxs, we also require that the types of the elements are in φ via the field φ ix.

5.2.1 Contexts

The context type is defined generically on the pattern functor of φ. We thus reuse the type family PF defined in Section 3. We have to distinguish between different type constructors that make up the pattern functor, and therefore define Ctx as a datatype family:

```
data family Ctx f :: *φ → *φ → *
```

Like the context stack, a context frame is parameterized over both the type of the resulting index and the type of the hole.

The simple cases are for constant types, sums and products. There is a correspondence between the context of a datatype and its formal derivative (McBride 2001):

```
data instance Ctx (K a)   ix b = CK Void
data instance Ctx (f :+: g) ix b = CL (Ctx f ix b)
                                   | CR (Ctx g ix b)
data instance Ctx (f :×: g) ix b = CI (Ctx f ix b) (g I* ix)
                                   | C2 (f I* ix) (Ctx g ix b)
```

For constants, there are no recursive positions, hence we produce an empty datatype, i.e., a datatype with no constructors:

```
data Void
```

For a sum, we are given either an f or a g, and compute the context of that. For a product, we can descend either left or right. If we descend into f, we pair a context for f with g. If we descend into g, we pair f with a context for g.

We are left with the cases for I and (:>:). According to the analogy with the derivative, the context of the identity should be the unit type. However, we are in a situation where there are multiple types involved. The type index of I fixes the type of the hole. We express this type equality as follows, by means of a GADT:¹

```
data instance Ctx (I xi) ix b where
  CId :: Ctx (I xi) ix xi
```

For the case of tags, we have a similar situation. A tag does not affect the structure of the context, it only provides information for the type system. In this case, not the type of the hole, but the type of the context itself is required to match the type index of the tag:

```
data instance Ctx (f :>: xi) ix b where
  CTag :: Ctx f xi b → Ctx (f :>: xi) xi b
```

This completes the definition of Ctx. We can convince ourselves that instantiating Ctx to PF AST results in a datatype that is isomorphic to Ctx_{AST}. It is also quite a bit more complex than the hand-written variant, but fortunately, the programmer never has to use it directly. Instead, we can interface with it using generic navigation functions.

5.2.2 Navigation

The navigation functions are again generically defined on the structure of the pattern functor. Thus, we define them in a class Zipper:

```
class Zipper φ f where
  ...
```

We will fill this class with methods incrementally.

Down To move down in a tree, we define a generic function *first* in our class Zipper:

```
class Zipper φ f where
  ...
  first :: (∀b. φ b → b → Ctx f ix b → a) →
          f I* ix → Maybe a
```

The function takes a functor f I_{*} ix and tries to split off its first recursive component. This is of some type b where we know φ b. The rest is a context of type Ctx f ix b. The function takes a continuation parameter that describes what to do with the two parts. Function *down* is defined in terms of *first*:

```
down :: Loc φ ix → Maybe (Loc φ ix)
down (Loc p x cs) =
  first (λp' z c → Loc p' z (Cons p c cs)) (from p x)
```

We try to split the tree in focus x. If this succeeds, we get a new focus z and a new context frame c. We push c on the stack.

¹ Currently, GHC does not allow instances of datatype families to be defined as GADTs. In the actual implementation, we therefore simulate the GADT by including an explicit proof of type equality (Peyton Jones et al. 2006; Baars and Swierstra 2002).

We define *first* by induction on the structure of pattern functors. Constant types constitute the leaves in the tree. We cannot descend, and return *Nothing*.

```
instance Zipper φ (K a) where
  ...
  first f (K a) = Nothing
```

In a sum, we descend further, and add the corresponding context constructor *CL* or *CR* to the context.

```
instance (Zipper φ f, Zipper φ g) ⇒ Zipper φ (f :+: g) where
  ...
  first f (L x) = first (λp z c → f p z (CL c)) x
  first f (R y) = first (λp z c → f p z (CR c)) y
```

We want to get to the first child. Therefore, we first try to descend to the left in a product. Only if that fails (*mplus*), we try to split the right component.

```
instance (Zipper φ f, Zipper φ g) ⇒ Zipper φ (f :×: g) where
  ...
  first f (x :×: y) = first (λp z c → f p z (C1 c y)) x
  'mplus' first (λp z c → f p z (C2 x c)) y
```

In the *l* case, we have exactly one possibility. We split $I(I_* x)$ into *x* and the context *CId* and pass the two parts to the continuation *f*:

```
instance El φ xi ⇒ Zipper φ (l xi) where
  ...
  first f (I (I_* x)) = return (f proof x CId)
```

It is interesting to see why this is type correct: the type of *x* is *xi*, so applying *f* to *x* instantiates *b* to *xi* and forces the final argument of *f* to be of type $Ctx (l xi) ix xi$. But that is exactly the type of *CId*.

Finally, for a tag, we also descend further and apply *CTag* to the context.

```
instance Zipper φ f ⇒ Zipper φ (f ▷: xi) where
  ...
  first f (Tag x) = first (λp z c → f p z (CTag c)) x
```

This is type correct because *Tag* introduces the refinement that *CTag* requires: applying *CTag* to *c* results in $Ctx (f ▷: xi) xi b$. This can be passed to *f* only if *ix* from the type of *first* is equal to *xi*. But it is, because the pattern match on *Tag* forces it to be.

Up Now that we can move down, we also want to move up again. We employ the same scheme as before: using an inductively defined generic helper function *fill*, we then define *up*. The function *fill* has the following type:

```
class Zipper φ f where
  ...
  fill :: φ b → b → Ctx f ix b → f l_* ix
```

The function takes a value together with a compatible context frame and plugs them together, producing a value of the pattern functor. This operation is total, so no *Maybe* is required in the result.

With *fill*, we can define *up* as follows:

```
up :: Loc φ ix → Maybe (Loc φ ix)
up (Loc p x Nil) = Nothing
up (Loc p x (Cons p' c cs)) = Just (Loc p' (to p' (fill p x c)) cs)
```

We cannot move up in the root of the tree and thus fail on an empty context stack. Otherwise, we pick the topmost context frame, and call *fill*. Since *fill* results in a value of the pattern functor, we have to convert back into the original form using *to*.

We start the definition of *fill* with the case for *K*. As an argument to *fill*, we need a context for *K*, for which we defined but one constructor *CK* with a *Void* parameter. In other words, in order to call *fill* on *K*, we have to produce a value of *Void*, which, apart

from \perp , is impossible. In the context of our *Zipper* library, we can guarantee that \perp is never produced for *Void*. We therefore define:

```
instance Zipper φ (K a) where
  ...
  fill p x (CK void) = impossible void
impossible :: Void → a
impossible void = error "impossible"
```

The definition of *fill* is very straight-forward: for *l*, we return the element to plug itself; for $(▷:)$ and $(:×:)$, we call *fill* recursively. In the case for products, we recurse into the context:

```
instance (Zipper φ f, Zipper φ g) ⇒ Zipper φ (f :×: g) where
  ...
  fill p x (C1 c y) = fill p x c :×: y
  fill p y (C2 x c) = x :×: fill p y c
```

Right As a final example of a navigation function, we define *right*. We again employ the same scheme as before. We define a generic function *next* with the following type:

```
class Zipper φ f where
  ...
  next :: (∀b. φ b → b → Ctx f ix b → a) →
         (φ b → b → Ctx f ix b → Maybe a)
```

The function takes a context frame and an element that fits into the context. By looking at the context, it tries to move the focus one element to the right, thereby producing a new element – possibly of different type – and a new compatible context. These can, as in *first*, be combined using the passed continuation.

With *next*, we can define *right*:

```
right :: Loc φ ix → Maybe (Loc φ ix)
right (Loc p x Nil) = Nothing
right (Loc p x (Cons p' c cs)) =
  next (λp z c' → Loc p z (Cons p' c' cs)) p x c
```

We cannot move right in the root of the tree, thus *right* fails in an empty context. Otherwise, we only need to look at the topmost context frame, and pass it to *next*, together with the current focus. On success, we take the new focus, and push the new context frame back on the stack.

Most cases of *next* are without surprises: calling *next* for *K* is again impossible; in sums and on tags we recurse. Since an *l* indicates a single child – a leaf in the tree – we cannot move right from there and return *Nothing*.

The most interesting case is the case for products. If we are currently in the first component, we try to move to the next element there, but if this fails, we have to select the first child of the second component, calling *first*. In that case, we also have to plug the old focus *x* back into its context *c*, using *fill*. If, however, we are already in the right component, we do not need a case distinction and just try to move further to the right using *next*.

```
instance (Zipper φ f, Zipper φ g) ⇒ Zipper φ (f :×: g) where
  ...
  next f p x (C1 c y) =
    next (λp' z c' → f p' z (C1 c' y)) p x c
    'mplus' first (λp' z c' → f p' z (C2 (fill p x c) c')) y
  next f p y (C2 x c) =
    next (λp' z c' → f p' z (C2 x c')) p y c
```

5.2.3 Using the Zipper

The functions *enter*, *leave* and *update* can be converted from the specific case for AST almost without change. The code is exactly as before, we only have to adapt the types.

```

enter :: (Family  $\varphi$ , Zipper  $\varphi$  (PF  $\varphi$ ))  $\Rightarrow$   $\varphi$  ix  $\rightarrow$  ix  $\rightarrow$  Loc  $\varphi$  ix
enter p x = Loc p x Nil
leave :: Loc  $\varphi$  ix  $\rightarrow$  ix
leave (Loc p x Nil) = x
leave loc = leave (fromJust (up loc))
update :: ( $\forall$ ix.  $\varphi$  ix  $\rightarrow$  ix  $\rightarrow$  ix)  $\rightarrow$  Loc  $\varphi$  ix  $\rightarrow$  Loc  $\varphi$  ix
update f (Loc p x cs) = Loc p (f p x) cs

```

Let us repeat the example from before, but now use the generic Zipper: apart from the additional argument to *enter*, nothing changes

```

enter Expr >>> down >>> down >>> right >>> update solve >>>
leave >>> return $ example

```

and the result is also the same:

```

Just (Let ("x" := Const 42) (Add (EVar "x") (EVar "y")))

```

6. Generic rewriting

Term rewriting can be specified generically, for arbitrary regular datatypes, if these are viewed as fixed points of functors (Jansson and Jeurig 2000, Van Noort et al. 2008). In the following we show how to generalize term rewriting even further, to work on families with an arbitrary number of datatypes. For reasons of space, we do not discuss generic rewriting in complete detail, but focus on the operation of matching the left-hand side of a rule with a term.

6.1 Schemes of regular datatypes

Before tackling matching on families of mutually recursive datatypes, we briefly sketch the ideas behind its implementation on regular datatypes. Consider how to implement matching for the simple version of the Expr datatype introduced in Section 2. First, we define expression schemes, which extend expressions with a constructor for rule meta-variables. Then we define matching of those schemes against expressions:

```

data ExprS = MetaVar String | ConstS Int
           | AddS ExprS ExprS | MulS ExprS ExprS
match :: ExprS  $\rightarrow$  Expr  $\rightarrow$  Maybe [(String, Expr)]

```

On success, *match* returns a substitution mapping meta-variables to matched subterms. For example, the call

```

match (MulS (MetaVar "x") (MetaVar "y"))
      (Mul (Const 6) (Const 9))

```

yields *Just* [(*"x"*, *Const 6*), (*"y"*, *Const 9*)].

To implement *match* generically, we need to define the scheme of a datatype generically. To this end, recall that a regular datatype is isomorphic to the type *Fix f*, for a suitably defined *f*. A meta-variable can appear deep inside a scheme, this suggests that the extension with *MetaVar* should take place inside the recursion, and hence on *f*. This motivates the following definition for schemes of regular datatypes:

```

type Scheme a = Fix (K String :+: PF a)

```

For example, the expression scheme that is used above as the first argument to *match* can be represented by

```

In (R (R (R (I (In (L (K "x"))))  $\times$  : I (In (L (K "y"))))))

```

6.2 Schemes of a datatype family and substitutions

A family of mutually recursive datatypes requires as many sorts of meta-variables as there are datatypes. For example, for the family used in Section 3, we need three meta-variables, ranging over Expr, Decl and Var, respectively. Fortunately, we can deal with all these meta-variables in one go:

```

type Scheme  $\varphi$  = HFix (K String :+: PF  $\varphi$ )

```

As in the regular case, the pattern functor is extended with a meta-variable representation. We want meta-variable representations to be polymorphic, so, unlike other constructors, *K String* is not tagged with (*▷*). Now, the same representation can be used to encode meta-variables that match, for example, Expr, Decl and Var.

Dealing with multiple datatypes affects the types of substitutions. We cannot use a homogeneous list of mappings as we did earlier, because different meta-variables may map to different datatypes. We get around this difficulty by existentially quantifying over the type of the matched datatype:

```

data DynIx  $\varphi$  =  $\forall$ ix. DynIx ( $\varphi$  ix) ix
type Subst  $\varphi$  = [(String, DynIx  $\varphi$ )]

```

6.3 Generic matching

Generic matching is defined as follows:

```

type MatchM s a = StateT (Subst s) Maybe a
matchM :: (Family  $\varphi$ , HZip  $\varphi$  (PF  $\varphi$ ))  $\Rightarrow$   $\varphi$  ix  $\rightarrow$ 
        Scheme  $\varphi$  ix  $\rightarrow$  I* ix  $\rightarrow$  MatchM  $\varphi$  ()
matchM p (HIn (L (K metavar))) (I* e)
    = do subst  $\leftarrow$  get
      case lookup metavar subst of
        Nothing  $\rightarrow$  put ((metavar, DynIx p e) : subst)
        Just _  $\rightarrow$  fail ("repeated use: " ++ metavar)
matchM p (HIn (R r)) (I* e)
    = combine matchM r (from p e)

```

Generic matching tries to match a term of type *I* ix* against a scheme of corresponding type *Scheme φ ix*. The resulting information is returned in the *MatchM* monad. The definition of *MatchM* uses *Maybe* for indicating possible failure, and on top of that monad we use the state transformer *StateT*. The state monad is used to thread the substitution as we traverse the scheme and the term in parallel. The class *HZip*, which contains functionality for zipping, is introduced in the following subsection.

Generic matching consists of two cases. When dealing with a meta-variable, we first check that there is no previous mapping for it. (For the sake of brevity, we do not show how to deal with multiple occurrences of a meta-variable.) If that is the case, we update the state with the new mapping. The second case deals with matching constructors against constructors. More specifically, this corresponds to matching *Mul (Const 6) (Const 9)* against *MulS (MetaVar "x") (MetaVar "y")*. This is handled by the generic function *combine*, which matches the two pattern functor representations. If the representations match (as in our example), then *matchM* is applied to the recursive occurrences (for instance, on *MetaVar "x"* and *Const 6*, and *MetaVar "y"* and *Const 9*).

Now we can write the following wrapper on *matchM* to hide the use of the state monad that threads the substitution:

```

match :: (Family  $\varphi$ , HZip  $\varphi$  (PF  $\varphi$ ))  $\Rightarrow$   $\varphi$  ix  $\rightarrow$ 
        Scheme  $\varphi$  ix  $\rightarrow$  ix  $\rightarrow$  Maybe (Subst  $\varphi$ )
match p scheme tm = execStateT (matchM p scheme (I* tm)) []

```

6.4 Generic zip and combine

The generic function *combine* is defined in terms of another function, which is a generalization of *zipWith* for arbitrary functors. Like *hmap*, the function *hzipM* is defined by induction on the pattern functor by means of a type class:

```

class HZip  $\varphi$  f where
  hzipM :: Monad m  $\Rightarrow$ 
    ( $\forall$ ix.  $\varphi$  ix  $\rightarrow$  r ix  $\rightarrow$  r' ix  $\rightarrow$  m (r'' ix))  $\rightarrow$ 
    f r ix  $\rightarrow$  f r' ix  $\rightarrow$  m (f r'' ix)

```

The function *hzipM* takes an argument that combines the *r* and *r'* structures stored in the pattern functor. The traversal is performed in a monad to notify failure when the functor arguments do not match, and to allow the argument to use state, for example.

In the case of *combine*, we are not interested in the resulting merged structure (*r' ix*). Indeed, *matchM* stores information only in the state monad, so we define *combine* to ignore the result.

```
data K* a b = K* {unK* :: a}
combine :: (Monad m, HZip  $\phi$  f) =>
  ( $\forall ix. \phi ix \rightarrow r ix \rightarrow r' ix \rightarrow m ()$ ) ->
  f r ix -> f r' ix -> m ()
combine f x y = do hzipM wrapf x y
  return ()
where wrapf p x y = do fp x y
  return (K* ())
```

In the above, K_* is used to ignore the type *ix* in the result. The definition of *hzipM* does not differ much from that used when dealing with a single regular datatype:

```
instance El  $\phi$  xi => HZip  $\phi$  (I xi) where
  hzipM f (I x) (I y) = liftM I (f proof x y)
instance (HZip  $\phi$  a, HZip  $\phi$  b) => HZip  $\phi$  (a : $\times$ : b) where
  hzipM f (x1 : $\times$ : x2) (y1 : $\times$ : y2)
    = liftM2 (: $\times$ :) (hzipM f x1 y1) (hzipM f x2 y2)
instance (HZip  $\phi$  a, HZip  $\phi$  b) => HZip  $\phi$  (a : $+$ : b) where
  hzipM f (L x) (L y) = liftM L (hzipM f x y)
  hzipM f (R x) (R y) = liftM R (hzipM f x y)
  hzipM f _ _ = fail "zip failed in :+:"
```

```
instance HZip  $\phi$  f => HZip  $\phi$  (f : $\triangleright$ : ix) where
  hzipM f (Tag x) (Tag y) = liftM Tag (hzipM f x y)
instance Eq a => HZip  $\phi$  (K a) where
  hzipM f (K x) (K y) | x == y = return (K x)
  | otherwise = fail "zip failed in K"
```

In the definition above, we use *liftM* and *liftM2* to turn the pure structure constructors into monadic functions.

7. Related work

Malcolm (1990) shows how to define two mutually recursive types as initial objects of functor-algebras. Swierstra et al. (1999) show how to implement fixed points for mutual recursive datatypes in Haskell. They introduce a new fixed point for every arity of mutually recursive datatypes. None of these approaches can be used as a basis for an implementation of fixed points for mutually recursive datatypes in Haskell suitable for implementing generic programs. Higher-order fixed points like our HFix have been used by Bird and Paterson (1999) and Johann and Ghani (2007) to model folds on nested datatypes.

Several authors discuss how to generate folds and other recursive schemes on mutually recursive datatypes (Böhm and Berarducci 1985; Sheard and Fegaras 1993; Swierstra et al. 1999; Lämmel et al. 2000). Again, the definitions in these papers cannot be directly generalised to families of arbitrary many datatypes in Haskell.

Mitchell and Runciman (2007) show how to obtain traversals for mutually recursive datatypes using the class *Biplate*. However, the type on which an action is performed remains fixed during a traversal. In contrast, the recursion schemes from Section 4.4 can apply their function arguments to subtrees of different types.

Since dependently typed programming languages have a much more powerful type system than Haskell extended with GADTs and type families, it is possible to define fixed-points for mutually

recursive datatypes in many dependently typed programming languages. Benke et al. (2003) give a formal construction for mutually recursive datatypes as indexed inductive definitions in Alfa. Some similarities with our work are that the pattern functor argument is indexed by the datatype sort, and recursive positions specify the sort index of the subtree. Altenkirch and McBride (2003) show how to do generic programming in the dependently typed programming language OLEG. We believe that it is easier to write generic programs on mutually recursive datatypes in our approach, since we do not have to deal with kind-indexed definitions, environments, type applications, datatype variables and argument variables, in addition to the cases for sums, products and constants.

McBride (2001) first described a generic Zipper on regular datatypes, which was implemented in Epigram by Morris et al. (2006). The Zipper has been used as an example of a type-indexed datatype in Generic Haskell (Hinze et al. 2004), but again only for regular datatypes. The dissection operator introduced by McBride (2008) is also only defined for regular datatypes, although McBride remarks that an implementation in a dependently typed programming language for mutually recursive datatypes is possible.

In most other generic programming approaches, recursion is not explicitly represented and types occurring at such positions cannot be changed. Such approaches cannot be used to define the examples in this paper. For example SYB (Lämmel and Peyton Jones 2003) and EMGM (Oliveira et al. 2006) cannot be used to define *fold*. In *fold*, the algebra function is applied to constructor fields where the recursive positions have been replaced by the carrier of the algebra. However, in these approaches, values at recursive positions cannot be transformed into values of other types. PolyLib (Norell and Jansson 2004) supports the definition of *fold*, but it is limited to regular datatypes. A similar argument can be made for type-indexed types. Generic approaches that support type-indexed types (Oliveira and Gibbons 2005) need to represent recursion explicitly in order to define the Zipper and generic rewriting.

8. Conclusions

Until now, many powerful generic algorithms were known, but their adoption in practice has been hindered by their restriction to regular datatypes. In this paper, we have shown that we can overcome this restriction in a way that is directly applicable in practice: using recent extensions of Haskell, we can define generic programs that exploit the recursive structure of datatypes on families of arbitrarily many mutually recursive datatypes. For instance, extensive use of generic programming becomes finally feasible for compilers, which are often based on an abstract syntax that consists of many mutually recursive datatypes. Furthermore, our approach is non-invasive: the definitions of large families of datatypes need not be modified in order to use generic programming.

Additionally, we have demonstrated our approach by implementing several recursion schemes such as *compos* and *fold*, the Zipper, and rewriting functionality.

Based on this paper, the libraries *multirec* and *zipper* have been developed. They are available for download from HackageDB. A version of the rewriting library based on *multirec* will be released soon.

The *multirec* library contains Template Haskell code to automatically generate the boilerplate code for a family of datatypes. Also, in addition to the functionality shown here, the library offers access to the names of constructors. Furthermore, in this paper we have focused on functions that consume or transform values rather than functions that produce values. This, however, is no limitation, and we have for instance implemented a function that generates values of a particular type according to certain criteria using *multirec*.

Initial measurements indicate that our approach is in the slower half of generic programming libraries for Haskell. On the other

hand, some benchmarked functions expose significant optimisation opportunities to GHC. In one such example, GHC optimisations removed the intermediate use of generic values leaving code that is similar to what one would write manually. As a result, this example runs only 1.7 times slower than handwritten code (compared to 68.7 for the slowest library). More details can be found in the thesis of Rodriguez (2009).

In its current form, our approach cannot be used directly on parameterized types and does not support functor composition. We have, however, prototypical code that demonstrates that our approach can be extended to support these concepts without too much difficulty, and we plan to integrate this functionality into the library in the near future.

We plan to study the application of our representation using ($\text{:}\triangleright\text{:}$) to arbitrary GADTs, hopefully giving us *fold* and other generic operations on GADTs, as in the work of Johann and Ghani (2008).

In parallel to the Haskell version, we have also experimented with an Agda (Norell 2007) version of our library, using dependent types. The Agda version has proved to be invaluable in thinking about the development without having to worry about Haskell limitations at the same time. As to Haskell, we hope that the support for type families, which we rely on very much, will continue to stabilize in the future, and that perhaps the kind system will be slightly improved with possibilities to encode kinds such as our $\ast\phi$, or with the possibility to define kind synonyms.

Acknowledgements José Pedro Magalhães and Marcos Viera commented on a previous version of this paper. Claus Reinke suggested to us the “type families desugaring trick” to get around a problem with the type checker in an older version of GHC. The anonymous reviewers have provided several useful suggestions for improving the presentation. This research has been partially funded by the Netherlands Organisation for Scientific Research (NWO), through its projects on “Real-life Datatype-Generic Programming” (612.063.613) and “Scriptable Compilers” (612.063.406).

References

- T. Altenkirch and C. McBride. Generic programming within dependently typed programming. In *Generic Programming*, pages 1–20. Kluwer, 2003.
- A. Baars and D. Swierstra. Typing dynamic typing. In *ICFP’02*, pages 157–166, 2002.
- M. Benke, P. Dybjer, and P. Jansson. Universes for generic programs and proofs in dependent type theory. *Nordic J. of Comp.*, 10(4):265–289, 2003.
- R. Bird and R. Paterson. Generalised folds for nested datatypes. *Formal Aspects of Computing*, 11:11–2, 1999.
- C. Böhm and A. Berarducci. Automatic synthesis of typed Λ -programs on term algebras. *Theoretical Computer Science*, 39:135–154, 1985.
- B. Bringert and A. Ranta. A pattern for almost compositional functions. In *ICFP’06*, pages 216–226, 2006.
- J. Cheney and R. Hinze. A lightweight implementation of generics and dynamics. In *ACM SIGPLAN Haskell Workshop*, 2002.
- J. Gibbons. Generic downwards accumulations. *SCP*, 37(1–3):37–65, 2000.
- R. Hinze. A new approach to generic functional programming. In *POPL’00*, pages 119–132, 2000a.
- R. Hinze. Polytypic values possess polykinded types. In *MPC’00*, volume 1837 of *LNCSS*, pages 2–27. Springer, 2000b.
- R. Hinze. Generics for the masses. In *ICFP’04*, pages 236–243, 2004.
- R. Hinze, J. Jeuring, and A. Löh. Type-indexed data types. *SCP*, 51(2):117–151, 2004.
- S. Holdermans, J. Jeuring, A. Löh, and A. Rodriguez. Generic views on data types. In *MPC’06*, volume 4014 of *LNCSS*, pages 209–234. Springer, 2006.
- G. Huet. The Zipper. *JFP*, 7(5):549–554, 1997.
- P. Jansson and J. Jeuring. A framework for polytypic programming on terms, with an application to rewriting. In *WGP’00*, 2000.
- P. Jansson and J. Jeuring. PolyP — a polytypic programming language extension. In *POPL’97*, pages 470–482, 1997.
- P. Jansson and J. Jeuring. Polytypic unification. *JFP*, 8(5):527–536, 1998.
- J. Jeuring. Polytypic pattern matching. In *FPCA’95*, pages 238–248, 1995.
- P. Johann and N. Ghani. Foundations for structured programming with GADTs. In *POPL’08*, pages 297–308, 2008.
- P. Johann and N. Ghani. Initial algebra semantics is enough! In *Proceedings, Typed Lambda Calculus and Applications*, pages 207–222, 2007.
- R. Lämmel and S. Peyton Jones. Scrap your boilerplate: A practical design pattern for generic programming. pages 26–37. ACM Press, 2003.
- R. Lämmel, J. Visser, and J. Kort. Dealing with large bananas. In *WGP’00*, 2000.
- A. Löh. *Exploring Generic Haskell*. PhD thesis, Utrecht University, 2004.
- G. Malcolm. Data structures and program transformation. *SCP*, 14:255–279, 1990.
- C. McBride. Clowns to the left of me, jokers to the right (pearl): dissecting data structures. In *POPL’08*, pages 287–295, 2008.
- C. McBride. The derivative of a regular type is its type of one-hole contexts. strictlypositive.org/diff.pdf, 2001.
- E. Meijer, M. Fokkinga, and R. Paterson. Functional programming with bananas, lenses, envelopes, and barbed wire. In *FPCA’91*, volume 523 of *LNCSS*, pages 124–144. Springer, 1991.
- N. Mitchell and C. Runciman. Uniform boilerplate and list processing. In *ACM SIGPLAN Haskell Workshop*, 2007.
- P. Morris, T. Altenkirch, and C. McBride. Exploring the regular tree types. In *Types for Proofs and Programs*, LNCSS. Springer, 2006.
- T. van Noort, A. Rodriguez, S. Holdermans, J. Jeuring, and B. Heeren. A lightweight approach to datatype-generic rewriting. In *WGP’08*, 2008.
- U. Norell. *Towards a practical programming language based on dependent type theory*. PhD thesis, Chalmers University of Technology, 2007.
- U. Norell and P. Jansson. Polytypic programming in Haskell. In *IFL’03*, volume 3145 of *LNCSS*, pages 168–184. Springer, 2004.
- B. C. d. S. Oliveira and J. Gibbons. Typecase: A design pattern for type-indexed functions. In *ACM SIGPLAN Haskell Workshop*, 2005.
- B. C. d. S. Oliveira, R. Hinze, and A. Löh. Extensible and modular generics for the masses. In H. Nilsson, editor, *TFP’06*, pages 199–216, 2006.
- S. Peyton Jones, editor. *Haskell 98 Language and Libraries: The Revised Report*. Cambridge University Press, Cambridge, 2003.
- S. Peyton Jones, D. Vytiniotis, S. Weirich, and G. Washburn. Simple unification-based type inference for GADTs. In *ICFP’06*, pages 50–61, 2006.
- A. Rodriguez. *Towards Getting Generic Programming Ready for Prime Time*. PhD thesis, Utrecht University, 2009.
- T. Schrijvers, S. Peyton Jones, M. Chakravarty, and M. Sulzmann. Type checking with open type functions. In *ICFP’08*, pages 51–62, 2008.
- T. Sheard and L. Fegaras. A fold for all seasons. In *FPCA’93*, pages 233–242, 1993.
- T. Sheard and S. Peyton Jones. Template meta-programming in Haskell. In *ACM SIGPLAN Haskell Workshop*, 2002.
- D. Swierstra, P. Azero, and J. Saraiva. Designing and implementing combinator languages. In *AFP*, volume 1608 of *LNCSS*, pages 150–206. Springer, 1999.